

Calcul explicite de la courbure des tissus calibrés ordinaires :

J.P. Dufour et D. Lehmann

1 Introduction

Notons

$c(n, h) := \frac{(n-1+h)!}{(n-1)!h!}$ la dimension de l'espace vectoriel des polynômes homogènes de degré h à coefficients complexes, en n variables,

d et n deux entiers ($n \geq 2$, $d > n$),

k_0 l'entier tel que $c(n, k_0) \leq d < c(n, k_0 + 1)$.

Nous avons défini dans [CL] les tissus holomorphes de codimension un dits *ordinaires* (cette terminologie fait allusion au fait que cette propriété est génériquement vérifiée localement). Nous avons démontré

- que le rang d'un tel d -tissu sur une variété complexe M de dimension n était majoré par l'entier $\pi'(n, d) = \sum_{h=1}^{k_0} (d - c(n, h))$ (strictement inférieur, pour $n \geq 3$, au nombre de Castelnuovo $\pi(n, d)$ qui est le genre arithmétique maximum des courbes algébriques irréductibles de degré d dans l'espace projectif complexe \mathbb{P}_n de dimension n),

- que, si d était égal à $c(n, k_0)$ (nous dirons alors que le tissu est *calibré*), les relations abéliennes du tissu au dessus du complémentaire $M' = M \setminus S$ d'un certain sous-ensemble analytique S de dimension au plus $n-1$, s'identifiaient aux sections holomorphes s d'un certain fibré holomorphe $\mathcal{E} \rightarrow M'$ de rang $\pi'(n, d)$, dont la dérivée covariante ∇s , relativement à une certaine connexion holomorphe naturelle ∇ sur \mathcal{E} , était nulle.

[Dans le cas $n = 2$ des tissus planaires, S est vide, tous les tissus sont à la fois calibrés ($d = c(2, d-1)$) et ordinaires (non-nullité des déterminants de Vandermonde). En outre, $\pi'(2, d) = \pi(2, d)$. La connexion dont il est question a alors été définie par A. Hénaut ([H1]) quel que soit d . Sa courbure généralise celle donnée par Blaschke-Dubourdieu ([B]) pour $d = 3$, et cette généralisation à tout d pour $n = 2$ avait été esquissée par Pantazzi ([P]), évidemment dans un autre langage que celui des connexions, qui n'étaient pas définies à cette époque.]

La nullité de la courbure de cette connexion permet donc de montrer que le rang du tissu est maximum, c'est-à-dire égal au rang $\pi'(n, d)$ du fibré \mathcal{E} , sans qu'il soit besoin d'exhiber les relations abéliennes et de montrer leur indépendance linéaire. Malheureusement, cette courbure est difficilement calculable "à la main", en dehors de cas très élémentaires (même le cas du 5-tissu de Bol en dimension deux nécessite en pratique l'utilisation d'un ordinateur).

Nous nous étions contentés, dans [CL], de montrer l'existence de cette connexion, en esquissant une méthode de calcul, que nous avons essayé de mettre en oeuvre sur ordinateur avec L. Flaminio et Y. Hantout, mais qui n'a pas abouti avec eux. Depuis, nous avons réussi à la faire fonctionner, mais nous proposons dans cet article une autre méthode, plus simple, nécessitant des temps de calcul beaucoup plus courts, avec programmation¹ explicite des calculs correspondants sur MAPLE. Nous expliquerons plus loin la différence entre ces deux méthodes.

¹Un programme existerait déjà, rédigé par O. Ripoll, mais uniquement dans le cas $n = 2$ des tissus planaires, et qui n'a pas été publié à notre connaissance.

Parmi les premiers figurent en particulier le tissu de Bol et les $c(n, 4)$ -tissus de Pereira-Pirio ($k_0 = 4$) qui le généralisent, notés $\mathcal{W}(A_{0,n+3})$ dans [Pe]. Ce sont les tissus dont les intégrales premières sont les birapports de toutes les familles de 4 nombres (pris dans un ordre arbitraire) parmi les $(n+3)$ nombres $(x_1, \dots, x_n, 0, 1, \infty)$ supposés distincts dans la droite projective complexe. Il a été démontré que ces tissus étaient ordinaires et de rang maximum, par Bol pour $n = 2$, Pereira-Pirio pour $n = 3$, Pereira pour n quelconque ([P]). Nous avons rédigé une feuille de calcul dans laquelle il suffit de préciser n , et avons effectivement fait tourner le programme pour $n = 2, 3, 4$ et 5.

Pour répondre négativement à une conjecture de Chern-Griffiths selon laquelle, parmi les relations abéliennes des tissus exceptionnels, il y en avait nécessairement qui faisaient intervenir des polylogarithmes, Pirio a exhibé ([Pi]) 6 (resp. 10) relations abéliennes purement algébriques et linéairement indépendantes des 5-tissus planaires $(x, y, x + y, x - y, x^2 + y^2$ ou $xy)$ (resp. du 6-tissu planaire $(x, y, x + y, x - y, x^2 + y^2, xy)$). Nous avons évidemment vérifié que la courbure de ces tissus était nulle, ainsi d'ailleurs que celle du 7-tissu $(x, y, x + y, x - y, x^2 + y^2, x^2 - y^2, xy)$ de rang 15. Mais, l'ayant constatée par ordinateur pour $n = 3$ ou 4, nous avons pu en déduire pour tout n la nullité de la courbure des $c(n, 4)$ tissus \mathcal{WB}_n du même type définis par les fonctions

et l'une des familles de fonctions

(voir la démonstration dans la section 7).

On pourrait vouloir remédier à ce défaut en laissant MAPLE choisir lui-même les variables principales. Mais on s'aperçoit alors qu'il fait ce choix de façon plus ou moins aléatoire, variant d'une fois à l'autre, et que l'on ne contrôle plus rien du tout. En outre, le résultat n'est alors lisible qu'en cas de courbure nulle, puisque la trivialisatation par rapport à laquelle est calculée la forme de courbure varie ; or cette courbure est toujours un invariant intéressant du tissu : en particulier, l'existence d'un sous-fibré de \mathcal{E} , invariant par la connexion, et sur lequel la courbure est nulle, permet de borner inférieurement le rang du tissu par le rang de ce sous-fibré (on observe immédiatement par exemple que le 8-tissu planaire $(x, y, x + y, x - y, x^2 + y^2, x^2 - y^2, xy, x^4 + y^4)$ n'est pas de rang maximum 21, mais qu'il est au moins de rang 19).

Un d -tissu sur M n'est en fait qu'un feuilletage $\tilde{\mathcal{F}}$ sur l'espace total d'un certain revêtement de M à d -feuilletés. Sur un ouvert U de M au dessus duquel ce revêtement est trivial, il revient au même de se donner d feuilletages \mathcal{F}_i sur M : on dit alors que le tissu est *totalemt décomposable* au dessus de U .

²Nous ne sommes pas experts en MAPLE, et notre programme n'est peut-être pas le plus astucieux qu'on puisse imaginer ; mais il a au moins le mérite d'exister et de fonctionner.

décomposable. Mais le fibré \mathcal{E} dans le cas des tissus ordinaires, et la connexion tautologique dont on le munit dans le cas des tissus qui sont en plus calibrés, sont en fait définis globalement sur tout l'ouvert $M' = M \setminus S$.

On suppose donc le tissu défini par la donnée de d feuilletages holomorphes \mathcal{F}_i de codimension 1 sur la variété complexe M de dimension n , ($d > n$), en position générale au moins faible³. Quitte à remplacer M par un ouvert, on supposera désormais le tissu régulier sur tout M .

Une *relation abélienne* sur un ouvert U (supposé connexe et simplement connexe) de M est alors la donnée d'une famille $(F_i)_i$ de fonctions holomorphes sur U , $1 \leq i \leq d$, telles que

- pour tout i , F_i est une intégrale première de \mathcal{F}_i (avec éventuellement des singularités),
- la somme $\sum_{i=1}^d F_i$ est une fonction constante sur U .

Ces intégrales premières n'étant définies qu'à une constante additive près, cela revient en fait à ne définir que leur différentielle $\omega_i = dF_i$, de sorte que l'on peut encore définir une relation abélienne comme une famille $(\omega_i)_i$, $1 \leq i \leq d$, de 1-formes holomorphes ω_i sur U (admettant éventuellement des singularités), qui sont

- (i) toutes fermées (donc localement exactes) : $d\omega_i = 0$,
- (ii) qui vérifient $T\mathcal{F}_i \subset \text{Ker } \omega_i$ quel que soit i ($T\mathcal{F}_i = \text{Ker } \omega_i$ en tout point où ω_i n'est pas nulle),
- (iii) telle que $\sum_{i=1}^d \omega_i = 0$.

Dans le but d'introduire (\mathcal{E}, ∇) , nous allons en donner une définition équivalente en termes d'opérateurs différentiels. Notons $T\mathcal{F}_i$ le sous-fibré de TM formé des vecteurs tangents à \mathcal{F}_i , et A_i le sous-fibré de T^*M formé des formes linéaires nulles sur $T\mathcal{F}_i$ (dual du fibré quotient $TM/T\mathcal{F}_i$).

Soit $Tr : \oplus_{i=1}^d A_i \rightarrow T^*M$ l'homomorphisme de fibrés vectoriels (appelé *Trace*), défini par

$$Tr((\omega_i)_i) = \sum_{i=1}^d \omega_i.$$

L'hypothèse de position générale au moins faible signifie qu'il est de rang maximum n : son noyau $A = \text{Ker } Tr$ est donc un fibré vectoriel de rang $d - n$.

On définit un opérateur différentiel linéaire $D : J^1 A \rightarrow B$ d'ordre 1, où $B = (\wedge^2 T^*M)^{\oplus d}$, en associant, à toute section $s = (\omega_i)_i$ de A , la famille $(d\omega_i)_i$ des différentielles. Une *relation abélienne* est alors une section holomorphe s de A telle que $D(j^1 s) = 0$.

Avec les notations précédentes, posons $R_1 = \text{Ker}(D : J^1 A \rightarrow B)$: c'est l'espace des *relations abéliennes formelles à l'ordre 1*. Plus généralement, l'espace des *relations abéliennes formelles à l'ordre h* est le noyau du $(h - 1)$ -ième prolongement D_h de l'opérateur différentiel D ($= D_1$) :

$$R_h = \text{Ker}(D_h : J^h A \rightarrow J^{h-1} B).$$

Notant $\pi_h : R_h \rightarrow R_{h-1}$ la restriction à R_h de la projection $J^h A \rightarrow J^{h-1} A$, nous montrerons que les éléments de R_h qui se projettent par π_h sur un élément donné a_{h-1} de R_{h-1} sont les solutions d'un système linéaire $\Sigma_h(a_{h-1})$ de $c(n, h + 1)$ équations à d inconnues, dont la partie homogène ne dépend pas de a_{h-1} .

3 Comparaison des deux méthodes

Supposons chaque feuilletage \mathcal{F}_i défini sur U par une intégrale première u_i sans singularité.

³ On dit que le tissu est en position générale faible (resp. forte) si, en tout point m de la partie régulière M_0 du tissu, il existe au moins n des feuilletages parmi les d dont les espaces tangents en m sont en position générale (resp. si toute famille de n feuilletages parmi les d a cette propriété).

Dans la première méthode, nous avons supposé les coordonnées locales $x = (x_1, \dots, x_n)$ choisies de façon que $\frac{\partial}{\partial x_n}$ soit transverse à tous les feuilletages \mathcal{F}_i ($1 \leq i \leq d$) ; ceux-ci pouvaient donc être définis par les 1-formes holomorphes intégrables

$$\eta_i = dx_n + \sum_{\alpha=1}^{n-1} p_{i\alpha}(z) dx_\alpha, \text{ où } p_{i\alpha} = \frac{(u_i)'_\alpha}{(u_i)'_n}.$$

Les relations abéliennes (ω_i) étaient définies par les familles (f_i) de fonctions holomorphes telles que $\omega_i = f_i \eta_i$, et les relations abéliennes formelles à l'ordre k en un point m étaient définies par la valeur des f_i et de leurs dérivées partielles jusqu'à l'ordre k en ce point. Mais nous réduisons à $d.(k+1)$ le nombre de ces inconnues en observant que toutes les dérivées partielles cherchées s'exprimaient à l'aide des seules dérivées partielles successives par rapport à x_n .

Maintenant, nous ne faisons plus jouer de rôle particulier à la coordonnée x_n , et cherchons les intégrales premières F_i sous la forme $F_i = G_i(u_i)$, où G_i désigne une fonction holomorphe d'une seule variable. Notant g_i la dérivée de G_i , on a maintenant $\omega_i = g_i(u_i) du_i$, et les relations abéliennes formelles à l'ordre k en un point m sont alors définies par la valeur des fonctions g_i et de leurs dérivées jusqu'à l'ordre k au point $u_i(m)$, de sorte que le nombre d'inconnues est encore égal à $d.(k+1)$. Mais, outre que les coordonnées locales jouent désormais toutes le même rôle, les calculs sont beaucoup plus simples et plus rapides.

4 Localisation et calculs

Notations

i désigne un indice variant de 1 à d ,

λ, μ, \dots des entiers variant de 1 à n .

$L = (\ell_1, \ell_2, \dots, \ell_n)$ désigne un multi-indice formé d'entiers $\ell_\lambda \geq 0$,
 $|L| := \sum_\lambda \ell_\lambda$ s'appelle le *degré* de L .

Si $L = (\ell_1, \ell_2, \dots, \ell_n)$, et $L' = (\ell'_1, \ell'_2, \dots, \ell'_n)$, $L + L' := (\ell_1 + \ell'_1, \dots, \ell_n + \ell'_n)$.

1_λ désigne le multi-indice obtenu avec 1 à la place λ et 0 ailleurs.

Lorsque $\ell_\lambda \geq 1$, $L - 1_\lambda$ désigne le multi-indice obtenu en remplaçant ℓ_λ par $\ell_\lambda - 1$.

Relativement à des coordonnées locales $x = (x_1, \dots, x_n)$ dans \mathbb{C}^n , on notera $\partial_\lambda a$ ou a'_λ la dérivée partielle $\frac{\partial a}{\partial x_\lambda}$ d'une fonction holomorphe a ou d'une matrice à coefficients holomorphes.

Plus généralement, $\partial_L a$ ou a'_L désigne la dérivée partielle $\frac{\partial^{|L|} a}{(\partial x_1)^{\ell_1} \dots (\partial x_n)^{\ell_n}}$ d'ordre $|L|$.

4.1 Principes du calcul

Se donner une intégrale première $F_i = G_i(u_i)$ à constante additive près revient maintenant à se donner la fonction dérivée $g_i = (G_i)'$. Chaque fibré A_i étant désormais trivialisé par du_i , on pose $\omega_i = g_i(u_i) du_i$ (une telle forme est nécessairement fermée). Se donner une relation abélienne revient alors à se donner une famille (g_i) de fonctions holomorphes d'une variable ($1 \leq i \leq d$) telles que $\sum_i g_i(u_i) du_i \equiv 0$, soit :

$$\sum_i g_i(u_i) (u_i)'_\lambda \equiv 0 \quad \text{pour tout } \lambda.$$

[Les fonctions u_i sont données ; les inconnues sont les fonctions g_i].

Soit R_h l'espace des relations abéliennes formelles à l'ordre h . Les dérivées partielles successives des relations (iii) vont permettre de calculer R_h .

4.2 Les coefficients $M_L^h(u)$

Lemme 1 : *Pour toute fonction holomorphe u de n variables, et toute fonction holomorphe g d'une variable,*

(i) *les dérivées $\left((g \circ u) u'_\lambda\right)'_L$, sont des combinaisons linéaires*

$$\left((g \circ u) u'_\lambda\right)'_L = \sum_{h=0}^{|L|} M_{L+1_\lambda}^h(u) \cdot (g^{(h)} \circ u)$$

des dérivées successives $g^{(h)}$ de g (on a posé $g^{(0)} = g$), dont les coefficients $M_{L'}^h(u) = M_{L+1_\lambda}^h(u)$ ne dépendent que de u et du multi-indice $L' = L + 1_\lambda$, et non de la décomposition de celui-ci sous la forme $L + 1_\lambda$.

(ii) *On les calcule par récurrence sur $|L|$ à l'aide des formules*

$$\begin{aligned} M_{1_\lambda}^0(u) &= u'_\lambda, \\ M_{L+1_\mu}^0(u) &= \partial_\mu M_L^0(u), \\ M_{L+1_\mu}^h(u) &= \partial_\mu M_L^h(u) + M_L^{h-1}(u) \cdot u'_\mu \quad \text{si } 1 \leq h \leq |L| - 1, \\ M_{L+1_\mu}^{|L|}(u) &= M_L^{|L|-1}(u) \cdot u'_\mu. \end{aligned}$$

En particulier, on obtient les formules

$$M_L^0(u) = \partial_L u, \quad \text{et} \quad M_L^{|L|-1}(u) = \prod_{\lambda=1}^n (u'_\lambda)^{\ell_\lambda} \quad \text{lorsque } L = (\ell_1, \ell_2, \dots, \ell_n).$$

Le lemme résulte immédiatement de ce que, la forme $d(G(u))$ est fermée, G désignant une primitive de g .

Pour un tissu défini localement par les intégrales premières u_i , on posera :

$$M_{i,L}^h = M_L^h(u_i).$$

4.3 Les équations E_L

Il résulte des considérations précédentes qu'un élément de R_h au dessus d'un point $m \in M$ est représenté par ses composantes $(j_{u_i(m)}^h g_i)_i \in \oplus_i J^h A_i$, et que chaque composante $j_{u_i(m)}^h g_i$ est entièrement définie par la famille des nombres $\left(w_i^{(k)} = (g_i^{(k)} \circ u_i)(m)\right)_{0 \leq k \leq h}$. Pour qu'une telle famille appartienne à R_h , il faut et il suffit que soient vérifiées toutes les équations (E_L) pour $1 \leq |L| \leq h+1$, définies de la façon suivante : si $L = (\ell_1, \dots, \ell_n)$ avec $|L| \geq 1$, on choisit un indice λ tel que $\ell_\lambda \geq 1$; (E_L) désigne alors l'équation

$$(E_L) \quad \sum_{i=1}^d \sum_{h=0}^{|L|-1} M_{i,L}^h \cdot w_i^{(h)} = 0,$$

qui ne dépend pas du choix de l'indice λ tel que $\ell_\lambda \geq 1$. On en déduit le

Théorème 2 : *Si un élément $a_{h-1} \in R_{h-1}$ est défini par les nombres $(w_i^{(k)})_{i,k}$, ($0 \leq k \leq h-1$), les éléments de R_h se projetant sur a_{h-1} sont définis par les d nombres $(w_i^{(h)})_i$ solutions du système linéaire $\Sigma_h(a_{h-1})$ des $c(n, h+1)$ équations*

$$(E_L) \quad \sum_i C_i^L \cdot w_i^{(h)} = - \sum_{i=1}^d \sum_{k=0}^{h-1} M_{iL}^k \cdot w_i^{(k)}, \quad (|L| = h+1)$$

à d inconnues $w_i^{(h)}$, ($1 \leq i \leq d$), où $C_i^L = \prod_{\lambda=1}^n ((u_i)'_\lambda)^{\ell_\lambda}$ lorsque $L = (\ell_1, \ell_2, \dots, \ell_n)$.

On définit, pour tout $1 \leq h \leq k_0$, les matrices $P_h = ((C_i^L))$, ($1 \leq i \leq d$, $|L| = h$), de taille $d \times c(n, h)$, dont les colonnes sont indexées par les indices $i \in \{1, \dots, d\}$, et les lignes par les multi-indices L de degré $|L| = h$, avec $C_i^L = \prod_{\lambda=1}^n ((u_i)'_\lambda)^{\ell_\lambda}$ lorsque $L = (\ell_1, \ell_2, \dots, \ell_n)$.

L'ensemble $\mathcal{P}(n, h)$ des multi-indices L de degré $|L| = h$ est ordonné de la façon suivante : la suite $L_1 \cdots L_{n-1}$ est l'écriture d'un entier naturel en base $h+1$: on ne conserve que ceux de ces entiers tels que $\sum_{i=1}^{n-1} L_i \leq h$, et l'on complète la suite L_1, \dots, L_{n-1} par $L_n := h - \sum_{i=1}^{n-1} L_i$: l'ensemble $\mathcal{P}(n, h)$ est maintenant identifié à un sous-ensemble de \mathbb{N} , que l'on munit de l'ordre induit.

Ceci permet d'écrire le système $\Sigma_h(a_{h-1})$ sous la forme matricielle

$$< P_{h+1}, w^{(h)} > = S(a_{h-1}),$$

où $w^{(h)}$ désigne la matrice colonne à d lignes $((w_i^{(h)}))_i$, et $S(a_{h-1})$ la matrice colonne à $c(n, h+1)$ lignes $((-\sum_{i=1}^d \sum_{k=0}^{h-1} M_{iL}^k \cdot w_i^{(k)}))_{|L|=h+1}$.

4.4 Majoration du rang des tissus ordinaires

Dire que le tissu est *ordinaire* signifie que tous les systèmes $\Sigma_h(a_{h-1})$ sont de rang maximum

$$\inf(d, c(n, h+1))$$

lorsque a_{h-1} se projette sur M en dehors d'un sous-ensemble analytique S de dimension au plus $n-1$ (ou éventuellement vide).

Notons k_0 l'entier tel que

$$c(n, k_0) \leq d < c(n, k_0 + 1),$$

Puisque $C_i^{L+1n} = (u_i)'_n C_i^L$, il suffit que les systèmes Σ_h soient de rang maximum $c(n, h+1)$ lorsque $h \leq k_0$, pour qu'ils soient encore de rang maximum d quel que soit $h > k_0$. Il revient donc encore au même de dire que le tissu est ordinaire si le système des équations E_L pour $|L| \leq k_0 - 1$ est de rang maximum $c(n+1, k_0 - 1) - 1$ et P_{k_0} de rang d .

On en déduit le

Théorème 3 : *Si le tissu est ordinaire, les solutions de $\Sigma_h(a_{h-1})$ forment un espace affine de dimension $d - c(n, h+1)$ pour $h \leq k_0 - 1$, et la restriction de R_h à M' est alors un fibré vectoriel de rang $\sum_{k=1}^{h+1} (d - c(n, k))$ au dessus de M' .*

Pour $h \geq k_0$, les systèmes $\Sigma_h(a_{h-1})$ ont 0 ou 1 solution, de sorte qu'un jet infini de relation abélienne formelle en un point $m \in M'$ (a fortiori un germe puisqu'on est dans un contexte analytique) est entièrement déterminé par sa projection sur R_{k_0-1} . On en déduit le

Théorème 4 ([CL]) : *Le rang d'un tissu ordinaire (c'est-à-dire le maximum de la dimension⁴ de l'espace des germes de relation abélienne en un point $m \in M$) est au plus égal au rang*

$$\pi'(n, d) := \sum_{k=1}^{k_0} (d - c(n, k))$$

du fibré $R_{k_0-1}|_{M'}$ (restriction de R_{k_0-1} à M').

En effet, en un point $m \in M'$, toutes les matrices P_h sont de rang maximum $c(n, h+1)$ pour $h \leq k_0 - 1$ et d pour $h \geq k_0$. L'espace affine des solutions de chaque système $\Sigma_h(a_{h-1})$ a donc la dimension $d - c(n, h+1)$ pour $h \leq k_0 - 1$. Il est de dimension 0 ou est vide pour $h \geq k_0$. On en déduit que le rang du tissu est majoré par $\pi'(n, d)$ en les points de M' ; un raisonnement élémentaire de semi-continuité prouve qu'il est aussi a fortiori majoré par $\pi'(n, d)$ en les points de S .

4.5 Connexion tautologique des tissus calibrés ordinaires

Si le tissu est calibré ($d = c(n, k_0)$), et ordinaire, on suppose désormais (quitte se restreindre à un ouvert partout dense) que tous les systèmes Σ_h sont de rang maximum pour $h \leq k_0 - 1$.

Posons $\mathcal{E} := R_{k_0-2}$: c'est un fibré vectoriel holomorphe de rang $\pi'(n, d) = \sum_{k=1}^{k_0-1} (d - c(n, k))$, la projection $R_{k_0-1} \rightarrow \mathcal{E}$ étant maintenant un *isomorphisme* de fibrés vectoriels. Notons $u : \mathcal{E} \xrightarrow{\cong} R_{k_0-1}$ l'isomorphisme inverse. Le fibré

$$R_{k_0-1} := J^1 \mathcal{E} \cap (J^{k_0-1} A)$$

est l'intersection des fibrés $J^1 \mathcal{E}$ et $J^{k_0-1} A$ dans $J^1(J^{k_0-2} A)$. Notant $\iota : R_{k_0-1} \subset J^1 \mathcal{E}$ l'inclusion naturelle, l'application composée $\iota \circ u : \mathcal{E} \rightarrow J^1 \mathcal{E}$ est une scission holomorphe de la suite exacte

$$0 \rightarrow T^* M' \otimes \mathcal{E} \rightarrow J^1 \mathcal{E} \xrightarrow{\iota \circ u} \mathcal{E} \rightarrow 0$$

et définit par conséquent une connexion holomorphe ∇ sur \mathcal{E} , que nous appellerons *la connexion tautologique*, dont la dérivation covariante associée est donnée par la formule

$$\nabla s = j^1 s - (\iota \circ u)(s).$$

Puisque $\iota \circ u$ se factorise à travers R_{k_0-1} , il est équivalent de dire, pour une section σ de A , que $j^{k_0-1} \sigma$ est une section de R_{k_0-1} ou que $\nabla(j^{k_0-2} \sigma)$ s'annule : les relations abéliennes au dessus de M' sont donc les sections holomorphes σ de A telles que $\nabla(j^{k_0-2} \sigma) = 0$.

Dire que la connexion tautologique est sans courbure équivaut alors à dire que le tissu est de rang maximum $\pi'(n, d)$ (le rang de \mathcal{E}) au voisinage de tout point de M . On a ainsi démontré :

Théorème 5 ([CL]) :

(i) *Pour les $c(n, k_0)$ -tissus ordinaires, les relations abéliennes s'identifient, par l'application $\sigma \rightarrow j^{k_0-2} \sigma$, aux sections holomorphes s de \mathcal{E} dont la dérivée covariante ∇s par rapport à la connexion tautologique est nulle.*

(ii) *Le tissu est alors de rang maximum $\pi'(n, d)$ ssi la courbure de la connexion tautologique est nulle.*

Expression explicite de la dérivation covariante :

Une section de \mathcal{E} est définie par la donnée des fonctions d'une variable $s_i^{(h)}$ pour $0 \leq h \leq k_0 - 2$, $i = 0 \cdots d$. Notons alors :

∇_λ la dérivée covariante par rapport à $\frac{\partial}{\partial x_\lambda}$,

⁴Si le tissu est en position générale forte, A. Hénaut ([H2]) a démontré que cette dimension ne dépendait pas de m .

et $(U_i(s))_{i=1,\dots,d}$ la solution du système cramérien $\Sigma_{k_0-1}(s)$. La définition de ∇ s'exprime alors par les formules :

$$\begin{aligned} (\nabla_\lambda s)_i^{(h)} \circ u_i &= \frac{\partial}{\partial z_\lambda} (s_i^{(h)} \circ u_i) - u'_\lambda \cdot (s_i^{(h+1)} \circ u_i) \quad \text{pour } h \leq k_0 - 3, \\ \text{et } (\nabla_\lambda s)_i^{(k_0-2)} \circ u_i &= \frac{\partial}{\partial z_\lambda} (s_i^{(k_0-2)} \circ u_i) - u'_\lambda \cdot U_i(s). \end{aligned}$$

Remarque : Les formules ci-dessus montrent que, localement, la connexion sur \mathcal{E} est la restriction d'une connexion définie sur un fibré trivial de rang $(k_0 - 1)d$; mais celui-ci n'a aucune signification intrinsèque, alors que \mathcal{E} et la connexion tautologique ont une signification intrinsèque globale, indépendante des coordonnées locales et du choix des intégrales premières locales u_i , le tissu n'ayant même pas à être totalement décomposable globalement.

Trivialisation de \mathcal{E} et forme de connexion :

Notons $\pi_h(s)$ la projection sur R_h d'une section s de \mathcal{E} pour $h \leq k_0 - 2$. La famille $s^{(h)} = (s_i^{(h)})_i$ appartient à l'ensemble des solutions de $\Sigma(\pi_{h-1}(s))$: il suffit donc de définir, pour tout $h \leq k_0 - 2$, un sous-ensemble I_h de $d - c(n, h + 1)$ entiers $i \in \{1, \dots, d\}$ pour lesquels la sous-matrice correspondante de P_{h+1} (c'est-à-dire celle dont les colonnes sont indexées par les indices $i \notin I_h$) a un déterminant non-nul⁵, pour en déduire une trivialisation locale de \mathcal{E} donnée par les sections $S_{i,h}$, ($i \in I_h$, $0 \leq h \leq k_0 - 2$), ainsi définies ; dans notre programmation, nous avons choisi pour I_h les $c(n, h + 1)$ derniers indices. C'est bien-entendu celà qui rend le programme sensible à l'ordre des u_i .

Pour $i \in I_h$, $h \leq k_0 - 2$,

$$(S_{i,h})_j^{(k)} \equiv 0 \text{ si } (j, k) \neq (i, h), \quad (S_{i,h})_i^{(h)} \equiv 1.$$

On peut en particulier calculer $\nabla(S_{i,h}) = \sum_\lambda (\nabla_\lambda(S_{i,h}) dz_\lambda)$, ($h \leq k_0 - 2$, $i \in I_h$), d'où la forme de connexion ω relative à cette trivialisation de \mathcal{E} .

La courbure :

On calcule alors la courbure

- soit par les crochets

$$K_{\lambda\mu} = [\nabla_\lambda, \nabla_\mu],$$

c'est-à-dire :

$$\langle K, s \rangle = \sum_{\lambda < \mu} \left(\nabla_\lambda(\nabla_\mu s) - \nabla_\mu(\nabla_\lambda s) \right) dz_\lambda \wedge dz_\mu,$$

- soit par la forme de courbure $\Omega = d\omega + \frac{1}{2}\omega \wedge \omega$ relative à la trivialisation précédente.

5 Programmation sur MAPLE 8

Exemple des tissus $\mathcal{W}(A_{0,n+3})$ de Pereira-Pirio (Bol pour $n = 2$), avec une déformation en G que l'on peut supprimer pour n grand afin de ne pas allonger exagérément les temps de calcul.

> restart;

Entrée des paramètres de base :

> n:= ... ; k0:=4 ;

⁵Attention : Il se peut qu'il faille encore restreindre l'ouvert de M' au dessus duquel on se place, car il n'est peut-être pas possible d'utiliser le même ensemble I_h en tous les points de M' .

Quelques entiers qu'on en déduit, qui seront utiles dans la suite (l'espace \mathcal{E} des $(k_0 - 2)$ jets de relation abéliennes sera un fibré de rang ro , inclus dans l'espace des vecteurs de dimension $alpha$; ses éléments seront le noyau de la matrice MM à $beta$ lignes et $alpha$ colonnes). L'entier ro est aussi égal au rang maximum $\pi'(n, d)$ du tissu.

```
> d:=binomial(n-1+k0,k0);
```

(cette condition exprime que le tissu est "calibré").

```
> alpha:=(k0-1)*d; ro:=k0*d-binomial(k0+n,k0)+1; beta:=alpha-ro;
```

```
> X:=seq(x[i],i=1..n);
```

Entrée des d intégrales premières (comme fonctions des $op(j, X)$, $j=1..n$) :

```
> apply(u,j,X):
```

```
> for j to n do u(j,X):=op(j,X) od;
```

```
> for j from 2 to n do for i to j-1 do u(n+i+binomial(j-1,2),X):=op(j,X)/op(i,X);
```

```
print(u(n+i+binomial(j-1,2))=(%)); od od;
```

```
> for j from 2 to n do for i to j-1 do
```

```
u(n+binomial(n,2)+i+binomial(j-1,2),X):=(op(j,X)-1+G)/(op(i,X)-1);
```

```
print(u(n+binomial(n,2)+i+binomial(j-1,2))=(%)); od od;
```

```
> for j from 2 to n do for i to j-1 do
```

```
u(n+2*binomial(n,2)+binomial(n,3)+i+binomial(j-1,2),X) :=
```

```
op(i,X)*(op(j,X)-1)/(op(j,X)*(op(i,X)-1));
```

```
print(u(n+2*binomial(n,2)+binomial(n,3)+i+binomial(j-1,2))=(%)); od od;
```

```
> for k from 3 to n do for j from 2 to k-1 do for i to j-1 do
```

```
u(n+2*binomial(n,2)+i+binomial(j-1,2)+binomial(k-1,3),X):=(op(i,X)-op(k,X))/(op(j,X)-op(k,X));
```

```
print(u(n+2*binomial(n,2)+i+binomial(j-1,2)+binomial(k-1,3))=(%)); od od od;
```

```
> for k from 3 to n do for j from 2 to k-1 do for i to j-1 do
```

```
u(n+3*binomial(n,2)+binomial(n,3)+i+binomial(j-1,2)+binomial(k-1,3),X) :=
```

```
op(j,X)*(op(i,X)-op(k,X))/(op(i,X)*(op(j,X)-op(k,X)));
```

```
print(u(n+3*binomial(n,2)+binomial(n,3)+i+binomial(j-1,2)+binomial(k-1,3))=(%)); od od od;
```

```
> for k from 3 to n do for j from 2 to k-1 do for i to j-1 do
```

```
u(n+3*binomial(n,2)+2*binomial(n,3)+i+binomial(j-1,2)+binomial(k-1,3),X):=
```

```
(op(j,X)-1)*(op(i,X)-op(k,X))/((op(i,X)-1)*(op(j,X)-op(k,X)));
```

```
print(u(n+3*binomial(n,2)+2*binomial(n,3)+i+binomial(j-1,2)+binomial(k-1,3))=(%)); od od od;
```

```
> for m from 4 to n do for k from 3 to m-1 do for j from 2 to k-1 do for i to j-1 do
```

```
u(n+3*binomial(n,2)+3*binomial(n,3)+i+binomial(j-1,2)+binomial(k-1,3)+binomial(m-1,4),X):=
```

```
(op(j,X)-op(m,X))*(op(i,X)-op(k,X))/((op(i,X)-op(m,X))*(op(j,X)-op(k,X)));
```

```
print(u(n+3*binomial(n,2)+3*binomial(n,3)+i+binomial(j-1,2)+binomial(k-1,3)+binomial(m-1,4))=(%));
```

```
od od od od;
```

A partir de maintenant, le programme ne dépend plus du tissu introduit.

Calcul des coefficients $M(j, h, L)$:

```
> with(LinearAlgebra):
```

```

> interface(rtablesize=(k0)*d);
> apply(M,j,h,L,X):
Calcul des premiers coefficients M(j,0,EE(i))
> apply(EE,i):
> apply(delta,t,s):
> for t to n do for s to n do if (t=s) then delta(t,s):=1 else delta(t,s):=0 end if od od;
> for i to n do EE(i):=[seq(delta(i,s),s=1..n)] od;

(EE(λ) est le multi-indice noté  $1_\lambda$  ci-dessus).
> for j to d do for i to n do M(j,0,EE(i),X):= diff(u(j,X),x[i]) ; > od od;
Génération et indexation des multi-indices de dérivation d'ordre 0 à  $k_0$ :
> apply(L,tau):apply(E,r,y):apply(LL,z):for l from 0 to  $k_0 * (k_0 + 1)^{(n-2)}$  do for k to  $k_0$  do
E(l,k):=Vector(n) od od;
> tau:=1:
> for k to  $k_0$  do for l from 0 to  $k * (k + 1)^{(n-2)}$  do p:=l: for s to n-1 do r:=p mod (k+1);E(l,k)[n-
s]:=r;p:=(p-r)/(k+1) od : SS:=sum('E(l,k)[u]', 'u'=1..n-1): if SS_i(k+1) then E(l,k)[n]:=k-SS:L(tau):=E(l,k):tau:=tau+1
end if od od;
> for t to binomial(n+k0,k0)-1 do LL(t):=[seq(L(t)[i],i=1..n)] od ;
> M(j,-1,L)=0 et M(j,h,L):=0 pour  $h > |L| - 1$ 
(où  $|L| = \sum (L[i], i = 1..n)$  :)
> for j to d do for tau to binomial(n+k0,k0)-1 do M(j,-1,LL(tau),X):=0 od od ;
> for j to d do for tau to binomial(n+k0,k0)-1 do SS:=sum('LL(tau)[i]', 'i'=1..n);for h from SS to
 $k_0$  do M(j,h,LL(tau),X):=0 od od od;
Calcul des  $M(j,h,L)$  par récurrence sur  $|L|$  :
> for j to d do for ss from 2 to  $k_0$  do for tau to binomial(n+k0,k0)-1 do if (sum('LL(tau)[i]', 'i'=1..n)=ss)
then for r to n do if (LL(tau)[r]=0) then else for h from 0 to ss-1 do
M(j,h,LL(tau),X):=
simplify(diff(M(j,h,LL(tau)-EE(r),X),x[r])+M(j,h-1,LL(tau)-EE(r),X)*diff(u(j,X),x[r])) od ;
r:=r+n fi od fi od od od;

On vérifie le résultat en les imprimant (on note provisoirement  $N(j,h,L) = M(j,h,L,X)$ ) ; cette
étape, qui peut utiliser du temps de calcul lorsque  $n$  et  $k_0$  augmentent, peut être supprimée.
> for j to d do for h from 0 to  $k_0$  do for tau to binomial(n+k0,k0)-1 do
print(N(j,h,LL(tau))=M(j,h,LL(tau),X)) od od od ;

Les matrices MM, QQ (matrices des systèmes d'équations  $E_L$  pour  $|L| \leq k_0 - 2$  et  $|L| = k_0 - 1$ 
et PP (matrice notée  $P_{k_0}$  ci-dessus) :
```

Le tissu est ordinaire si le rang de MM est beta et si celui de PP est d ; on ne vérifie pas directement la première condition, car on aura besoin d'explicitier ci-dessous une sous-matrice carrée YYY de MM, de rang beta.

Les lignes sont numérotées par l'indice tau de LL(tau). On numérote maintenant les colonnes :

```

> hh:=j- >floor((j-1)/d);ii:=j- >j-d*floor((j-1)/d):
> ff:=(tau,eta)- >M(ii(eta),hh(eta),LL(tau),X);
> MM:=simplify(Matrix(beta,alpha,ff));
> fff:=(tau,eta)- >M(ii(eta),hh(eta),LL(tau+ binomial(n+k0-1,k0-1)-1),X):
```

```

> QQ:=simplify(Matrix(d,(k0-1)*d,fff));
> fff:=(tau,eta)->M(eta,k0-1,LL(tau+ binomial(n+k0-1,k0-1)-1),X):
> PP:=Matrix(d,d,fff);
Calcul d'une base W(j) , j=1..ro, de l'espace des sections de  $\mathcal{E} := \text{Ker} (MM)$ 
Définition d'une sous-matrice carrée YYY de MM :
> Y(0):=MM:
> for k from 1 to (k0-1) do Y(k):=DeleteColumn (Y(k-1),(k0-k-1)*d+
binomial(k0-k+n-1,n-1)+1..(k0-k)*d) end do;
> YYY:=Y(k0-1);
> evalb(Rank(YYY)=beta);
(si le rang de YYY est strictement inférieur à beta, réessayer en modifiant l'ordre des u(i)).
> IYYY:=simplify(MatrixInverse(YYY));
> B(0):=MM:
> for k from 1 to (k0-1) do
B(k):=DeleteColumn(B(k-1),(k0-k-1)*d+1..(k0-k-1)*d+binomial(k0-k+n-1,n-1)) end do;
> B:=simplify(B(k0-1));
> for j from 1 to ro do ColB(j):=Column(B,j) end do;
> apply(a,j,s);
> for j from 1 to ro do for s from 1 to beta do a(j,s):=factor(factor
(simplify((simplify(-IYYY.ColB(j)))[s]))) ; print('a(' ,j,s , ')=', a(j,s)) end do end do;
Partition de la suite 1..(k0-1)d en R(h) et S(h) pour h de 0 à  $(k_0 - 2)d$  ;
nr(h) := nombre d'éléments dans R(h) ; ns(h) := nombre d'éléments dans S(h) :
> apply(R,h);apply(S,h);
> for h from 0 to k0-2 do R(h):=[seq(i,i=h*d+1..h*d+binomial(h+n,n-1))] end do;
> for h from 0 to k0-2 do nr(h):=binomial(h+n,n-1) end do;
> for h from 0 to k0-2 do S(h):=[seq(i,i=h*d+binomial(h+n,n-1)+1..(h+1)*d)] end do;
> for h from 0 to k0-2 do ns(h):=d-binomial(h+n,n-1) end do;
> VV:=proc(j) global V; V:=Vector((k0-1)*d);
> for i in R(0) do V[i]:=a(j,i) od;
> for h from 1 to (k0-2) do for i in R(h) do V[i]:=a(j,i-sum(ns(kkk),kkk=0..h-1)) od; od;
> for h from 0 to (k0-2) do for i in S(h) do
> if i= (j+sum(nr(kk),kk=0..h)) then V[i]:=1 else V[i]:=0 fi ;od;od;
> evalm(V): end proc ;
> apply(W,j);
> for j to ro do VV(j):W(j):=V od;
Matrice U exprimant les termes de rang  $k_0 - 1$  en fonction des termes de rang inférieur:
> evalb(Rank(PP)=d);
(deuxième condition pour que le tissu soit ordinaire)
> IPP:=MatrixInverse(PP);
> U:=simplify(-IPP.QQ);

```

```
> U:=simplify(U);
```

Définition de la connexion tautologique sur \mathcal{E} :

Expression des dérivées covariantes dans l'espace des vecteurs de dimension α :

```
> Nabla:=proc(VV,j) description "calcul du  $Nabla_{x(j)}$  du vecteur VV ; le résultat est le vecteur Vec";global Vec; Vec:= Vector((k0-1)*d);
```

```
> for h from 1 to (k0-2) do
```

```
> for i to d do
```

```
> Vec[i+(h-1)*d]:=diff(VV[i+(h-1)*d],x[j])- VV[i+h*d]*diff(u(i,X),x[j])od od;
```

```
> for i to d do Vec[i+(k0-2)*d]:=simplify(diff(VV[i+(k0-2)*d],x[j]) -(U.VV)[i]*diff(u(i,X),x[j])) od;
end proc;
```

Calcul de la forme de connexion relative à la trivialisation (W_j) :

```
> DerCov:=proc(i); "le résultat est la matrice A"; A:=Matrix(ro,ro);
```

```
> apply(DC,j,i):for j to ro do Nabla(W(j),i);DC(j,i):=Vec od:apply(f,i):f(i)=(s,j)-> DC(j,i)[s];
apply(N,i):N(i):=Matrix(alpha,ro,f(i));
```

```
> apply(Aa ,i,j):Aa(i,0):=N(i):for k to (k0-1) do
```

```
Aa(i,k):=DeleteRow(simplify(Aa(i,k-1)),(k0-k-1)*d+1..(k0-k-1)*d+binomial(k0-k+n-1,n-1)) end do;
```

```
A:=simplify(Aa(i,k0-1));end proc;
```

```
> apply(A,i):for i to n do A(i):=DerCov(i):print(connexion(i)=DerCov(i)) od;
```

connexion(i) est la composante sur dx_i de la forme de connexion relative à la trivialisation $(W(j))_j$.

Calcul de la forme de courbure K :

```
> apply(A,r,s): apply(f,r,s):
```

```
> for s to n do for r to n do f(r,s):=(i,j)->simplify(simplify(diff(A(r)[i,j],x[s]))):
```

```
> A(r,s):=Matrix(ro,ro,f(r,s)):od od:
```

```
> apply(AA,r,s):for s to n do for r to n do AA(r,s):=simplify(simplify(A(r).A(s))): od od:
```

```
> apply(ff,r,s):
```

```
> apply(K,r,s):
```

```
> for s from 2 to n do for r to s-1 do
```

```
ff(r,s):=(i,j)->simplify(simplify(A(r,s)[i,j]-A(s,r)[i,j]+AA(s,r)[i,j]-AA(r,s)[i,j]));
```

```
K(r,s):=Matrix(ro,ro,ff(r,s)):print(courbure(r,s)=K(r,s)) od od ;
```

Pour les tissus à paramètre G dont la courbure s'annule pour $G=0$, développement limité en G à l'ordre 0 de la courbure et le programme affiche les éléments en $O(G)$. (s'il n'y a pas de paramètre, le résultat est le même qu'à la ligne précédente). Attention à ce que la lettre G n'aie pas été utilisée comme symbole par ailleurs .

```
> apply(ffo,r,s):
```

```
> apply(Ko,r,s):
```

```
> for s from 2 to n do for r to s-1 do ffo(r,s):=(i,j)-> taylor(K(r,s)[i,j],G,1);
```

```
> Ko(r,s):=Matrix(ro,ro,ffo(r,s)):print(courbure(r,s)=Ko(r,s)) od od ;
```

courbure(r,s) est la composante sur $dx_r \wedge dx_s$ de la forme de courbure relative à la trivialisation $(W(j))_j$

6 Exemples tests :

Pour la satisfaction de l'oeil, afin de nous assurer que l'on n'obtenait pas systématiquement une courbure nulle à la suite d'une erreur de programmation, nous avons déformé certains des tissus de ces exemples à l'aide d'un paramètre scalaire G (le tissu concerné étant obtenu pour $G = 0$), et nous avons mis en évidence les coefficients non nuls de la courbure qui sont petits d'ordre G .

Pour $n = 2, 3, 4$, on note parfois dans cette section (x, y) , (x, y, z) , (x, y, z, t) les coordonnées locales.

1) $n = 2$, $k_0 = 2$: déformation du 3-tissu hexagonal avec développement limité en G de la courbure de Blaschke à l'ordre que l'on veut (temps de calcul : 07", rang 1) :

$$u_1 = x \quad u_2 = y \quad u_3 = x + y + Gf(x, y).$$

2) $n = 3$, $k_0 = 2$: l'un des rares exemples, en dehors du cas ($n = 2$, $k_0 = 2$) de la courbure de Blaschke, que l'on peut traiter sans ordinateur ; cf [CL]); (11", rang 3); on introduit une fonction holomorphe arbitraire $y \rightarrow F(y)$, $F \neq 25$:

$$u_1 = z \quad u_2 = x + y + z \quad u_3 = 2x + 4y + z \quad u_4 = 3x + 9y + z \quad u_5 = 4x + 16y + z \quad u_6 = 5x + F(y) + z.$$

3) Les tissus $\mathcal{W}(A_{0,n+3})$ de Pereira-Pirio ($n \geq 2$, $k_0 = 4$), avec une déformation en G que l'on peut supprimer à volonté, en particulier pour n ou k_0 grands, afin de ne pas allonger exagérément les temps de calcul : c'est l'exemple rédigé dans la section précédente.

$n = 2$ (tissu de Bol, rang 6) ; temps de calcul : 10",
 $n = 3$ (rang 26) ; temps de calcul : 30" sans déformation, 44" avec,
 $n = 4$ (rang 71) ; temps de calcul : 4'12" sans déformation, 12'02" avec,
 $n = 5$ (rang 150) ; temps de calcul : 55' sans déformation.

4) Généralisation \mathcal{WB}_n à tout n des 5-tissus planaires $(x, y, x + y, x - y, x^2 + y^2$ ou $xy)$ de Pirio, qui sont de rang maximum ($n \geq 2$, $k_0 = 4$) : voir la section suivante.

5) $n = 2$, $k_0 = 5$: déformation du 6-tissu $(x, y, x + y, x - y, x^2 + (1 + G)y^2, xy)$ de Pirio (11", rang 10) , ou du 7-tissu ($k_0 = 6$, 19", rang 15) obtenu en ajoutant encore $u_7 = x^2 - (1 + G)y^2$.

[quant au 8-tissu obtenu en ajoutant encore $u_8 = x^4 + y^4$ lorsque $G = 0$, on observe immédiatement que sa courbure n'est pas nulle, mais que le sous-fibré engendré par les 19 premiers vecteurs de la trivialisatation est préservé par la connexion et que la restriction de la courbure à ce sous-fibré est nulle (24") : il est donc de rang 19 ou 20].

6) $n = 2$, $k_0 = 8$: le 9-tissu exceptionnel de G. Robert (5'02" sans déformation, rang 28) :

$$x, y, \frac{x}{1+y}, \frac{1+x}{y}, \frac{x}{y}, \frac{1+x}{1+y}, \frac{y(1+x)}{x(1+y)}, \frac{(1+x)(1+y)}{xy}, \frac{x(1+x)}{y(1+y)}.$$

7 Les tissus \mathcal{WB}_n :

On observe que $c(n, 4)$ est égal à $n + 3\frac{n(n-1)}{2} + 3\frac{n(n-1)(n-2)}{6} + \frac{n(n-1)(n-2)(n-3)}{24}$ pour $n \geq 4$, à 15 pour $n = 3$ et à 5 pour $n = 2$.

Théorème 6 :

Les $c(n, 4)$ -tissus \mathcal{WB}_n admettant comme intégrales premières locales les fonctions

x_i ,
 $x_i + x_j$, $x_j - x_i$, $x_i.x_j$, ($i < j$),
 $x_i + x_j + x_k$, $x_i^2 + x_j^2 + x_k^2$, $x_i.x_j.x_k$, ($i < j < k$)
 $x_i.x_j.x_k.x_m$, ($i < j < k < m$)

sont ordinaires et de rang maximum $\pi'(n, c(n, 4))$.

Démonstration : Pour $n = 2$, le résultat est démontré par Pirio, qui explicite 6 relations abéliennes (algébriques) indépendantes. Pour $n = 3$ ou 4, on vérifie avec notre programme que la courbure est nulle (temps de calcul avec déformation : 24" pour $n = 3$, 3'19" pour $n = 4$). Puisque le 5-tissu planaire WB_2 de Pirio est de rang maximum 6, le tissu WB_n contient exactement $6 \frac{n!}{2!(n-2)!}$ relations abéliennes indépendantes ne faisant intervenir que 2 des n coordonnées locales. Puisque le 15-tissu WB_3 a une courbure nulle, c'est que le nombre de relations abéliennes indépendantes faisant intervenir les 3 coordonnées locales est égal à 8 ($= \pi'(3, 15) - 6 \frac{3!}{2!1!}$) ; par conséquent le nombre des relations abéliennes indépendantes de WB_n faisant intervenir exactement 3 des n coordonnées locales est égal à $8 \frac{n!}{3!(n-3)!}$. De même, puisque le 35-tissu WB_4 a une courbure nulle, c'est que le nombre de ses relations abéliennes indépendantes faisant intervenir les 4 coordonnées locales est égal à 3 ($= \pi'(4, 35) - 6 \frac{4!}{2!2!} - 8 \frac{4!}{3!1!}$). Il en résulte que, pour $n \geq 4$, WB_n possède au moins

$$6 \frac{n!}{2!(n-2)!} + 8 \frac{n!}{3!(n-3)!} + 3 \frac{n!}{4!(n-4)!}$$

relations relations abéliennes indépendantes (ne faisant intervenir que 2, 3 ou 4 variables). Or ce nombre est précisément égal à $\pi'(n, c(n, 4))$, comme on le vérifie aisément. Il suffit donc de démontrer que le tissu est ordinaire pour en déduire qu'il est de rang maximum.

Commençons par montrer que la matrice $P_4 (= PP)$ est inversible.

- On choisit un ordre arbitraire O2 sur les couples (i,j) tels que $i < j$, un ordre arbitraire O3 sur les triplets (i,j,k) tels que $i < j < k$ et un ordre arbitraire O4 sur les quadruplets (i,j,k,m) tels que $i < j < k < m$.

- On ordonne les colonnes (qui correspondent aux fonctions $u(i)$) en mettant d'abord les intégrales premières x_1, \dots, x_n puis celles qui font intervenir deux variables dans l'ordre O2, puis celles qui font intervenir trois variables dans l'ordre O3, puis les fonctions $x_i.x_j.x_k.x_m$ dans l'ordre O4.

- On réordonne maintenant les lignes (qui correspondent aux multi-indices L d'ordre 4). On met en premier ceux qui ne font intervenir qu'une variable $4_1, \dots, 4_n$ (j_i désignant le multi-indice dont tous les termes sont nuls sauf le i -ème égal à j) ; on range ensuite ceux qui font intervenir deux variables x_i et x_j en rangeant les triplets $(3_i+1_j, 2_i+2_j, 1_i+3_j)$ pour $i < j$ suivant O2 ; on range ensuite ceux qui font intervenir les trois variables x_i, x_j et x_k en ordonnant les triplets $(2_i+1_j+1_k, 1_i+2_j+1_k, 1_i+1_j+2_k)$ pour $i < j < k$ suivant O3 ; on range enfin les $1_i+1_j+1_k+1_m$ pour $i < j < k < m$ suivant O4.

Après ces ré-ordonnements P_4 devient une matrice par blocs, dont tous les blocs diagonaux sont inversibles, et les blocs sous les blocs diagonaux sont nuls.

Pour P_3 (resp. P_2 , resp. P_1), on agit de même, mais en ne conservant que la sous-matrice obtenue en éliminant les colonnes correspondant aux $x_i.x_j$, $x_i^2 + x_j^2 + x_k^2$, $x_i.x_j.x_k$ et $x_i.x_j.x_k.x_m$ (resp. en effaçant toutes les colonnes sauf celles qui correspondent aux x_i , $x_i + x_j$ et $x_i - x_j$, en ne gardant que la sous-matrice correspondant aux x_i), et les lignes correspondant aux multi-indices L d'ordre 3 (resp 2, resp 1)

Remarques :

(i) On aurait obtenu un résultat analogue en remplaçant les fonctions de 4 variables $x_i.x_j.x_k.x_m$ par $x_i + x_j + x_k + x_m$, ou par $(x_i)^2 + (x_j)^2 + (x_k)^2 + (x_m)^2$.

(ii) On n'a pas eu à utiliser le programme montrant que la courbure était nulle pour tout n , mais seulement pour $n = 3$ ou 4 .

(iii) La même méthode permet de redémontrer que les tissus $\mathcal{W}(A_{0,n+3})$ sont tous ordinaires de rang maximum $\pi'(n, c(n, 4))$.

Références :

[B] W. Blaschke et G. Bol, Geometrie der Gewebe, Die Grundlehren der Mathematik 49, Springer, 1938.

[Bo] G. Bol, Über ein bemerkenswertes Fünfgewebe in der Ebene, Abh. Math. Hamburg Univ., 11, 1936, 387-393.

[CL1] V. Cavalier et D. Lehmann, Ordinary holomorphic webs of codimension one, arXiv 0703596 v2[mathDS], 2007, et Ann. Sc. Norm. Super. Pisa, cl. Sci (5), vol XI (2012), 197-214.

[GR] G. Robert, Relations fonctionnelles polylogarithmiques et tissus plans, prépublication n 146, 2002, Université de Bordeaux I.

[H1] A. Hénaut, On planar web geometry through abelian relations and connections, Ann. of Math. 159 (2004), 425-445.

[Pa] A. Pantazi, Sur la détermination du rang d'un tissu plan, C.R. Acad. Sc. Roumanie 2, 1938, 108-111.

[Pe] J. V. Pereira, Resonance webs of hyperplane arrangements, Advanced studies in Pure Mathematics 99, 2010, 1-30.

[Pi] L. Pirio, Sur les tissus planaires de rang maximal et le problème de Chern, note aux CRAS, sér. I, 339 (2004), 131-136.

[PT] L. Pirio et J.M. Trépreau, Abelian functional equations, planar web geometry and polylogarithms, Selecta Mathematica, N.S., 11, n 3-4, 2005, 453-489.

Jean-Paul Dufour, ancien professeur à l'Université de Montpellier II,
1 rue du Portalet, 34820 Teyran, France
email : dufourh@netcourrier.com,

Daniel Lehmann, ancien professeur à l'Université de Montpellier II,
4 rue Becagrun, 30980 Saint Dionisy, France
email : lehm.dan@gmail.com,